

### Exercice 1. Terminaison

Montrer que les fonctions suivantes terminent, pour des arguments entiers :

```

1  let rec g a =
2      if a < 0 then 1
3      else if a mod 2 = 0 then g (a + 1)
4      else g (a - 3)

1  let rec f a b =
2      if a <= 0 || b <= 0 then 1
3      else if a mod 2 = 0 then f (a/3) (2*b)
4      else f (3*a) (b/5)

```

**Solution :**

- Supposons que l'on appelle **g** **a** et notons  $a_k$  les arguments de ses appels récursifs, avec  $a_0 = a$ .  
Soit  $k \in \mathbb{N}$ . Si  $a_k$  est pair,  $a_{k+1} = a_k + 1$  est impair et  $a_{k+2} = a_{k+1} - 3 = a_k - 2$  est pair. Ainsi,  $a_{k+2p}$  est une suite strictement décroissante d'entiers, qui devient donc négatif et la fonction termine.  
Si  $a_k$  est impair,  $a_{k+1}$  est pair et on se ramène au cas précédent.
- À chaque appel récursif de **f**, le produit des 2 arguments diminue strictement.  
Comme ce produit reste toujours entier, il devient donc négatif ou nul ce qui termine les appels récursifs.

### Exercice 2. Invariant de boucle simple

En utilisant un invariant de boucle, prouver que la fonction suivante renvoie bien la somme des éléments d'un tableau :

```

1  let somme t =
2      let s = ref 0 in
3      for i = 0 to Array.length t - 1 do
4          s := !s + t.(i)
5      done;
6      !s

```

On rappelle qu'un invariant de boucle est une propriété qui reste vraie à chaque itération de la boucle et qui permet de montrer que la fonction renvoie le bon résultat (ici, la somme des éléments de **t**).

On prouve cet invariant de boucle par récurrence sur le nombre d'itérations dans la boucle.

**Solution :** On pose  $H_i$  : "au début de l'itération  $i$  de la boucle **for**, **s** contient  $\sum_{k=0}^{i-1} t.(k)$ ".

$H_0$  est vrai car  $\sum_{k=0}^{i-1} t.(k) = 0 = s$  initialement.

Supposons  $H_i$ . Alors,  $s$  vaut  $\sum_{k=0}^{i-1} t.(k)$  au début de l'itération  $i$ .

Comme l'itération  $i$  ajoute  $t.(i)$  à  $s$ ,  $s$  contient  $\sum_{k=0}^i t.(k)$  au début de l'itération  $i + 1$ , ce qui prouve  $H_{i+1}$ .

Par principe de récurrence, la valeur renvoyée par **somme** est bien la somme des éléments de **t**.

### Exercice 3. Tranche maximum (algorithme de Kadane)

Soit **t** un tableau d'entiers. Une **somme consécutive** (ou tranche) dans **t** est de la forme  $\sum_{k=i}^j t.(k)$  (où  $i$  et  $j$  sont des indices de **t**). On note  $s$  la valeur maximum d'une somme consécutive.

1. Écrire une fonction **tranche\_max** prenant **t** en argument et renvoyant **s**, en complexité quadratique en la taille de **t**.

**Solution :**

```

1  let tranche_max t =
2      let maxi = ref t.(0) in
3      let n = Array.length t in
4      for i = 0 to n - 1 do
5          let sum = ref 0 in
6          for j = i to n - 1 do
7              sum := !sum + t.(j); (* sum contient la somme de t.(i) à t.(j) *)
8              maxi := max !maxi !sum
9          done;
10     done;
11     !maxi

```

Si  $j$  est un indice de  $\mathbf{t}$ , on note  $s_j$  la plus grande somme consécutive finissant en  $j$ . Dit autrement :

$$s_j = \max_{0 \leq i \leq j} \sum_{k=i}^j t.(k)$$

2. Calculer tous les  $s_j$ , si  $\mathbf{t} = [1; -4; 1; 5; -7; 0]$

**Solution :**

indice	0	1	2	3	4	5
$\mathbf{t}$	1	-4	1	5	-7	0
$\mathbf{s}$	1	-3	1	6	-1	0

3. Si  $j > 0$ , montrer que :

$$s_j = \max(s_{j-1} + t.(j), t.(j))$$

**Solution :** Soit  $i$  le début de la somme  $s_j$  (c'est à dire  $s_j = \sum_{k=i}^j t.(k)$ ).

- Si  $i = j$  : alors  $s_j = t.(j)$
- Sinon,  $i < j$  et  $s_j = \sum_{k=i}^{j-1} t.(k) + t.(j)$ . Clairement,  $\sum_{k=i}^{j-1} t.(k)$  est une tranche finissant en  $j - 1$ .

Supposons que  $\sum_{k=i}^{j-1} t.(k)$  ne soit pas une tranche maximum finissant en  $j - 1$  (c'est à dire  $\sum_{k=i}^{j-1} t.(k) < s_{j-1}$ ).

Alors  $s_{j-1} + t.(j)$  est une tranche finissant en  $j$  qui est plus grande que  $s_j$ . C'est absurde par définition de  $s_j$ .

Donc  $\sum_{k=i}^{j-1} t.(k) = s_{j-1}$  et  $s_j = s_{j-1} + t.(j)$ .

On a donc soit  $s_j = t.(j)$ , soit  $s_j = s_{j-1} + t.(j)$ . Comme  $s_j$  est la somme maximum, on conserve le plus grand des deux :

$$s_j = \max(s_{j-1} + t.(j), t.(j))$$

4. Comment peut-on exprimer  $\mathbf{s}$  en fonction de  $s_j$  ?

**Solution :**  $\mathbf{s}$  est le maximum des  $s_j$ .

5. En déduire une fonction `tranche_max` prenant  $\mathbf{t}$  en argument et renvoyant  $\mathbf{s}$ , en complexité linéaire en la taille de  $\mathbf{t}$ .

**Solution :**

```

1  let tranche_max t =
2    let s = ref t.(0) in
3    let sj = ref t.(0) in
4    for j = 1 to Array.length t - 1 do
5      sj := max (!sj + t.(j)) t.(j);
6      s := max !s !sj
7    done;
8    !s;;

```

6. Donner un invariant de boucle permettant de prouver que `tranche_max t` est correct.

**Solution :** Invariant de boucle  $H_j$  : "au début de l'itération  $j$  de la boucle `for`,  $sj$  est la tranche maximum finissant en  $t.(j)$  et  $s$  est le max des  $sk$  pour  $k \leq j$ "

7. Modifier votre fonction précédente pour obtenir les indices de début et fin de  $s$ .

**Solution :**

```

1  let tranche_max t = (* renvoie (tranche max, son début, sa fin *)
2    let s = ref t.(0) in
3    let sj = ref t.(0) in
4    let deb = ref 0 in (* indice du début de la somme dans s *)
5    let fin = ref 0 in (* indice de fin de la somme dans s *)
6    let deb_sj = ref 0 in (* indice du début de la somme dans sj *)
7    for j = 1 to Array.length t - 1 do (* O(n) où n = taille de t *)
8      sj := !sj + t.(j);
9      if !sj < t.(j) then (sj := t.(j); deb_sj := j);
10     if !s < !sj then (s := !sj; deb := !deb_sj; fin := j)
11   done;
12   (!s, !deb, !fin)

```

#### Exercice 4. Tri par insertion

1. Écrire une fonction `insere` telle que, si  $l$  est une liste triée et  $e$  un élément, `insere l e` renvoie une liste triée contenant  $e$  et les éléments de  $l$ .

**Solution :**

```

1  let rec insere e l = match l with
2    | [] -> [e]
3    | x::q -> if e < x then e::l
4              else x::insere e q

```

2. En déduire un algorithme de tri, en utilisant plusieurs fois `insere`. Prouver que ce tri est correct.

**Solution :**

```

1  let rec tri_insertion = function
2    | [] -> []
3    | e::q -> insere e (tri_insertion q)

```

3. Quelle est la complexité de ce tri ? Pourrait-on l'améliorer en utilisant une recherche par dichotomie pour `insere` ?

**Solution :** Si  $l$  est de taille  $n$ , `tri_insertion l` appelle  $n$  fois `insere` qui est en  $O(n)$ , d'où une complexité  $O(n^2)$ .

4. Réécrire `tri_insertion` avec un tableau au lieu d'une liste et en utilisant une recherche par dichotomie pour l'insertion. Que peut-on dire de sa complexité ? Et de son nombre de comparaisons ?

**Solution :** La recherche par dichotomie permet de trouver la position où insérer l'élément en  $O(\log(n))$  comparaisons. Comme on le répète  $n$  fois, on a un tri en  $O(n \log n)$  comparaisons.

Par contre, pour insérer l'élément, il faut déplacer tous les éléments suivants ce qui demande  $O(n)$  opérations. La complexité reste donc  $O(n^2)$ .

---

```
1  let swap t i j =
2    let tmp = t.(i) in
3    t.(i) <- t.(j);
4    t.(j) <- tmp;;
5
6  let insere t pos =
7    let rec dichot i j =
8      (* renvoie l'indice où t.(pos) doit être inséré dans t *)
9      if i >= j then i
10     else let m = (i + j)/2 in
11           if t.(m) < t.(pos) then dichot (m + 1) j
12           else dichot i m in
13    let k = dichot 0 pos in
14    for i = pos downto k+1 do (* on met t.(pos) à sa place *)
15      swap t (i-1) i
16    done;;
17
18 let tri_insertion t =
19   for i = 0 to Array.length t - 1 do
20     insere t i
21   done
```

---